Building a Database Application:
A Case Study: 
The application we will build is a simple CD Album Catalogue. It is a database application which stores basic information about a CD-Album collection, including Title, Artist Name, and associated image of the CD cover (if we have one). To create this we will create forms with toolbars, menus, list views, and more. We will include good practice in coding, and also, exception handling. As before, it makes good sense to print this entire tutorial, and implement it in VB.NET as you work through it. Also, make your own notes as you go to aid understanding.

Building the Interface:

Start by creating a new Windows application called Album Catalogue, and follow these initial steps:

1. Change the name of the default form to fclsMain.

2. Change the Text property of fclsMain to Album Catalogue.

3. Right-click the default form in Solution Explorer, choose Rename, and change the name of the file to fclsMain.vb
4. Set the Startup object to fclsMain

(i.e. Right-Click on “Album Catalogue” in the Solutions window, and select ‘Properties’ from the pop-up menu. Then, at the Text-box labelled ‘Startup object:, use the Drop-down menu and select the option, fclsMain. (See the next two screen shots)

Designing the Main Window

The interface for the main form will contain a List View control that lists the albums in the database. This List View will show the album title and artist’s name. In addition, it will display a picture of a camera if artwork is attached to the album information. The form will also have menus and a toolbar for accessing its functionality. 
Add an image list control to the form now, and name it imgMain. Before adding images to the image collection, open the TransparentColor property of the Image List control, and choose Silver on the web tab, as shown here.

Add the following images to the Image List’s collection.
(Click on imgMain to reveal its’ properties and click on the ellipsis ... at the collection option of the images property.)

 The images are supplied on the web. 
1. Open.bmp

2. New.bmp

3. Delete.bmp

4. Edit.bmp

5. Quit.bmp

6. Camera.bmp

Take care over the order of adding these images. The references 1 through 6, are referred to by number later, so, if you mix them up you will get some confusing results.
When you have done this, the Image Collection Editor should look like this:-

[image: image4.png]Image Collection Editor

SystemDrawing Btmap Propertis:

System Drawing.Btmap. [ElMise

System Drawing.itmap. HorizortaResalut 35

System Drawing Bitmap. PhysicalDinensior| {Width=16, Height=16

System Drawing Bitmap. PxeFormat |Formatébppindexed

System Drawing.Bitmap RawFormat_|Bmp

E3 16,16
VerticaResolution| 36





Building the Menu
1. Add a new Main Menu control to the form and name it mmuMain.

2. Create a new menu item on the menu bar by typing the text &File in the Type Here box. Change the name of the new item to mnuFile. Remember that the text you enter to create the menu item becomes the Text property value, not the name.

3. Click on the Type Here box below File, and enter the text &New Album. Change the Name (property) of this item to mnuNewAlbum. Set its Shortcut (property) to CtrlN.
(Use the drop-down menu at the Shortcut property).

4. Click the Type Here box below New Album, and enter the text &Delete Album. Change the name of this item to mnuDeleteAlbum.

5. Right-click the Type Here box immediately beneath the Delete Album item, and choose Insert Separator from the shortcut menu. See below. 

6. Click the Type Here box below the separator, and enter the text &Quit. Change the Name of the new item to mnuQuit and set its shortcut property to CtrlQ.

7. Click the Type Here box directly to the right of the File menu item, and enter the text &Edit. This creates a new top-level menu item. Name it mnuEdit
8. Click the Type Here box below Edit, and enter the text &Edit Album. Change the Name of the new item to mnuEditAlbum.

The menu is now complete.

Building the Toolbar

1. Add a new Toolbar control to the form and name it tbMain, and set its ImageList property to imgMain.

2. Using the Buttons collection (property), create the following toolbar buttons:-

Button 1:
Property

Value

​​​​​​​​​​​​
Name


tbbOpenDatabase

ImageIndex

0    (Use the drop-down menu to see the standard options)

Text


(make blank)

ToolTipText

Open Database
Button 2:

Property

Value
​​​​​​​​​​​​
Name


tbbSeparator1
Style


Separator (Again- Select from the drop-down menu)
Button 3:

Property

Value
​​​​​​​​​​​​
Name


tbbNewAlbum

ImageIndex

1


Text


(make blank)

ToolTipText

New Album

Button 4:

Property

Value
​​​​​​​​​​​​
Name


tbbEditAlbum

ImageIndex

3


Text


(make blank)

ToolTipText

Edit Album

Button 5:

Property

Value
​​​​​​​​​​​​
Name


tbbDeleteAlbum

ImageIndex

2


Text


(make blank)

ToolTipText

Delete Album

Your toolbar collection should now look like the one shown here:

Adding the List View to Display Albums

1. Add a new List View Control to the form and set its properties as follows:

Property

Value

Name


lvwAlbums
Anchor


Top. Left, Right, Bottom
FullRowSelect

True
HideSelection

False
Location

6,32
MultiSelect

False
Size


277,224

SmallImageList
imgMain
Sorting 

Ascending
View


Details
2. Add the following three columns to the Columns collection (property) of the List View control:-

Column 1:


Property

Value

Name


colAlbumName
Text


Album
Width


160

Column 2:


Property

Value


Name


colArtistName

Text


Artist

Width


100

Column 3

Property

Value


Name


colArtWorkFileName

Text


ArtWorkFileName

Width


0
*Note that specifying a column width of 0 specifies a hidden column.

The interface of your main form is now complete and it should look like the following:-

[image: image8.png]Ble Edt Vew Projct Buld Debug Data

B-a-sHd@ B o

Fomat  Tooks

» Debug

window

Tookox 2 x| Statrace felsMainvb [Design]* |

Data
Components
Windows Farms

X Ble  Edt
Painter

A ol HEERIN

Help

T,

A Uritabel Abun

Atst

Button
Texttox
Mairienu
Chedtgox
Radiobuttan
Groupgox
PictureBox.
Panel
Datacrid

EEOR G Q@E

ListBox
[ CheckeduistBox

Clipboard Ring -
General

B server el 3 Tookox

Task List - 0 Buld Errr tasks shown (fitered)
1] [l Description

imghain

B Mainttenut

Fie

Line

- BEERT-,

arx

[

Soltion Evplrer - Album 5 x
=2E 88
193 Solution ‘Album3 (1 project)
= 3 albums
- (@ References
9 assanblyinfo.sb
[ fesvan.sh

Resource Vien

" GA Solution Explrer

Properties 2 x
Tewlbums_System, Windows. Forms. Latien
4il=
ChedkBoes Fale -l
Columns (Collection)
ContextMenu (none)
Cusor Defat
Dock None —
Enabled e
Font Microsoft Sans Serf, 5.2¢
Forecolor B windowrert
FulRowselect True
Columns

The calumns shawn in Reprt view





Adding the Browse For Database
OpenFileDialog Control

You are going to let the user browse for a database at runtime, so add a new OpenFileDialog control, and set its properties as follows:-

Property

Value

Name

ofdChooseDatabase

Filter

Microsoft Access Databases (*.mdb
Title

Open Database
Designing the Album Dialogue Box

The secondary form you’ll create is the Album dialogue box, which will be used to create and edit albums in the database. Choose Add Windows Form from the project menu and create a new windows form with the name fclsAlbum.vb. )See next screen-shot). Set the form’s properties as follows:-

Click on Open and you get the following screen. Then set the properties (you should be getting good at this by now).

Property


Value

FormBorderStyle

FixedDialog
Icon


Album.ico
MaximizeBox

False
MinimizeBox

False
Size


298,264

Text


Album Maintenance
1. Add a new label control to the form and set its properties as follows:-
Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


lblTitle


Location


8,20
Size


40,16
Text


Title:




2. Add a new Text Box control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


txtTitle



Location


8,20

Size


40,16
Text


(make blank)




3. Add another label control to the form and set its properties as follows:-
Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


lblArtistName



Location


8,44
Size


40,16
Text


Artist:




4. Add another Text Box control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


txtArtistName



Location


56,40
Size


144,20
Text


(make blank)


5. Add a Picture Box control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


picArtWork



BorderStyle

FixedSingle
Location


56,72
Size


144,144

SizeMode


StretchImage




6. Add a Button control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


btnChooseCoverArt



Location


8,72
Size


40,23
Text


Art




7. Add another Button control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


btnOK



Location


208,16 
Size


75,23
Text


OK




8. Add a third Button control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


btnCancel



Location


208,46
Size


75,23
Text


Cancel




9. Change the AcceptButton property of the form to btnOK
10. Change the CancelButton property of the form to btnCancel
11. You’re going to let the user browse their hard-drives for artwork files, so add an OpenFileDialog control to the form and set its properties as follows:-

Property


Value
​​​​​​​​​​​​​​​​​​​​​​​​​​​Name


ofdChooseArtWork

Filter


Windows Bitmaps(*.BMP(JPEG Files(*.JPG


Title


ChooseArtWork


Your form should now look something like this:-

Writing the Code of the CD Cataloguer

Now that your forms are complete, it’s time to add the code behind them. As you know by now, this involves creating module-level variables, as well as code procedures and event procedures.

Writing the Code for the Main Window

1. Click the fclsMain.vb [Design] tab to view the main form in the designer.
2. Click the View Code button in the Solution Explorer to access the form’s code, and enter the following procedure:-

Private Sub ShowAlbums()

        Dim cnADONetConnection As New OleDb.OleDbConnection

        Dim daDataAdapter As New OleDb.OleDbDataAdapter

        Dim cbCommandBuilder As OleDb.OleDbCommandBuilder

        Dim dtAlbums As New DataTable

        Dim rowPosition As Integer = 0

        Try

            ' Open the database.

            cnADONetConnection.ConnectionString = _

                  "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & _

                  m_strDatabase

            cnADONetConnection.Open()

            ' Get all albums.

            daDataAdapter = _

New OleDb.OleDbDataAdapter("Select * From tblCDs",_ cnADONetConnection)

            daDataAdapter.Fill(dtAlbums)

            ' Set up the command builder for navigation.

            cbCommandBuilder = New OleDb.OleDbCommandBuilder(daDataAdapter)

            ' Clear the list.

            lvwAlbums.Items.Clear()

            ' Go through the recordset and show all of the CDs/albums

            Dim objListItem As ListViewItem

            Do Until rowPosition = dtAlbums.Rows.Count

                ' Create a new list item for the album.

                objListItem = lvwAlbums.Items.Add( _

                              dtAlbums.Rows(rowPosition)("Title").ToString, 0)

                objListItem.SubItems.Add( _

                  dtAlbums.Rows(rowPosition)("ArtistName").ToString)

                objListItem.SubItems.Add( _

                              dtAlbums.Rows(rowPosition)("ArtWorkFileName").ToString)

                ' If artwork is present, indicate this with an icon.

                If dtAlbums.Rows(rowPosition)("ArtWorkFileName").ToString.Length > 0 Then

                    ' Show the image of the camera to denote there is 

                    ' an image for this album.

                    objListItem.ImageIndex = 5

                Else

                    objListItem.ImageIndex = -1

                End If

                rowPosition = rowPosition + 1

            Loop

            ' Close the connection.

            cnADONetConnection.Close()

        Catch ex As Exception

            MessageBox.Show("An error has occurred! Error: " & ex.Message, _

                            "CD Catalog", MessageBoxButtons.OK, _

                            MessageBoxIcon.Error)

        End Try

    End Sub
3. Enter the following statement immediately below the Inherits statement at the top of the module to create a module-level variable, This variable will hold the path and filename of the open database.
Private m_strDatabase As String

4. Place the cursor after the EndSub statement for the ShowAlbums() procedure and press enter a few times . Next, enter the following procedure:
Private Sub NewAlbum()

        Try

           ' Make sure a database is loaded.

           If m_strDatabase = "" Then

            MessageBox.Show("No database has been opened.", "CDCatalog", _                     MessageBoxButtons.OK, MessageBoxIcon.Exclamation)

                Exit Sub

            End If

            ' Load the album maintenance form and let it know

            ' the user is entering a new album

            Dim frmAlbum As New fclsAlbum

            frmAlbum.Database = m_strDatabase

            frmAlbum.ShowDialog()

            ' Refresh the album list.

            Call ShowAlbums()

        Catch ex As Exception

            ' Raise the error to the user, preventing a crash

            ' to the desktop.

            MessageBox.Show("An error has occurred! Error: " & ex.Message, _

                            "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

    End Sub
5. Place the cursor after the End Sub statement for the procedure you just created, and press enter a few times. Next enter the following procedure:
Private Sub EditAlbum()

        Try

            ' Make sure an album has been selected.

            If lvwAlbums.SelectedItems.Count = 0 Then

                MessageBox.Show("There is no album selected.", _

                                "CD Catalog", MessageBoxButtons.OK, _

                                MessageBoxIcon.Information)

                Exit Sub

            End If

            ' Load the album maintenance form and let it know

            ' which album is being edited.

            Dim frmAlbum As New fclsAlbum

            frmAlbum.Database = m_strDatabase

            frmAlbum.ShowAlbum(lvwAlbums.SelectedItems(0).Text, _

                   lvwAlbums.SelectedItems(0).SubItems(1).Text, _

                   lvwAlbums.SelectedItems(0).SubItems(2).Text)

            frmAlbum.ShowDialog()

            ' Refresh the album list.

            Call ShowAlbums()

        Catch ex As Exception

            MessageBox.Show("An error has occurred! Error: " & ex.Message, _

                            "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

    End Sub

6. Place the cursor after the End Sub statement for the Edit Album  procedure you just created, and press enter a few times. Next enter the following procedure:
Private Sub DeleteAlbum()

        Dim cnADONetConnection As New OleDb.OleDbConnection

        Dim daDataAdapter As New OleDb.OleDbDataAdapter

        Dim dtAlbums As New DataTable

        Dim cbCommandBuilder As OleDb.OleDbCommandBuilder

        Try

            ' Make sure an album has been selected.

            If lvwAlbums.SelectedItems.Count = 0 Then

                MessageBox.Show("There is no album selected.", _

             "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Information)

                Exit Sub

            End If

            ' Open the database.

            cnADONetConnection.ConnectionString = _

                     "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & _

                     m_strDatabase

            cnADONetConnection.Open()

            ' Find the album in the database.

            daDataAdapter = _

            New OleDb.OleDbDataAdapter("SELECT * From tblCDs " & _

          "WHERE Title = '" & lvwAlbums.SelectedItems(0).Text & "' AND " & _

          "ArtistName = '" & lvwAlbums.SelectedItems(0).SubItems(1).Text & _

          "'", cnADONetConnection)

            daDataAdapter.Fill(dtAlbums)

            ' Set up the command builder so we can delete a record.

            cbCommandBuilder = New OleDb.OleDbCommandBuilder(daDataAdapter)

            ' Delete the album.

            dtAlbums.Rows(0).Delete()

            ' Commit the changes to the actual data source.

            daDataAdapter.Update(dtAlbums)

            ' Remove the album from the list.

            lvwAlbums.Items.Remove(lvwAlbums.SelectedItems(0))

        Catch ex As Exception

        MessageBox.Show("An error has occurred! Error: " & ex.Message, _

                   "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

    End Sub

7. Click fclsMain.vb [Design] to return to design view, Open the File Menu and double-click the New Album menu item. Add the following code to its Click event:

NewAlbum()

8. Click fclsMain.vb [Design] to return to design view, Open the Edit Menu and double-click the Edit Album menu item. Add the following code to its Click event:

Call EditAlbum()

9. Click fclsMain.vb [Design] to return to design view, Open the File Menu and double-click the Delete Album menu item. Add the following code to its Click event:

Call DeleteAlbum()

10. Click fclsMain.vb [Design] to return to design view, Open the File Menu and double-click the Quit Album menu item. Add the following code to its Click event:

Me.Close()

11. Next you’ll add the code for the toolbar. Click the fclsMain.vb [Design] to return to design view. Double-click the Toolbar control and add the following code to the ButtonClick event:

Try

            If e.Button Is tbbOpenDatabase Then

                Call OpenDatabase()

            ElseIf e.Button Is tbbNewAlbum Then

                Call NewAlbum()

            ElseIf e.Button Is tbbDeleteAlbum Then

                Call DeleteAlbum()

            ElseIf e.Button Is tbbEditAlbum Then

                Call EditAlbum()

            End If

        Catch ex As Exception

            MessageBox.Show("An error has occurred! Error: " _

& ex.Message, "CD Catalog", _

            MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try
12. Create the following new procedure in fclsMain.vb

      Private Sub OpenDatabase()

        Try

            ' Let the user browse for and select a database.

            ofdChooseDatabase.ShowDialog()

            If ofdChooseDatabase.FileName <> "" Then

                m_strDatabase = ofdChooseDatabase.FileName

                Call ShowAlbums()

            End If

        
Catch ex As Exception

            
MessageBox.Show("An error has occurred! Error: _ 

" & ex.Message, "CD Catalog", _

            
MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

End Sub
13. The last procedure you’ll create in fclsMain.vb will enable a user to double-click an album in the list to edit it. Click fclsMain[Design} again to display the form in Design view and then double-click the List View to access its default event.  Choose DoubleClick from the event drop-down list and enter the following code:
Call EditAlbum()

Writing Code for the Album Maintenance Dialogue Box:
The Album Maintenance form (fclsAlbum), is used to add and edit albums. It has one method, Show Album, which is called by the main form to initiate editing an album. In addition, it has a property entitled Database that will accept the full path and filename of the chosen database.
The Show Album method will accept the information for the album being edited so that it knows which record to edit, and won’t have to look up the information in the database. 

1. Click the fclsAlbum.vb [Design] tab to show the Album Maintenance form.

2. Click the View Code button in the Solution Explorer to access the code behind the form. Add the following below the Inherits statement to create three module- level variables

Private m_strAlbumTitle As String

    

Private m_strArtistName As String

    

Private m_strArtWorkFileName As String

   Private m_strDatabase As String

3. Add the following property procedure (right below the Windows Form Designer generated code box);

Public Property Database() As String

        Get

            Database = m_strDatabase

        End Get

        Set(ByVal Value As String)

            m_strDatabase = Value

        End Set

   
 End Property
When steps 2 and 3 are done your code window should look like this:-

4. Next add the following new procedure which is used to display an album for editing:

Public Sub ShowAlbum(ByVal strAlbumTitle As String, _

                     ByVal strArtistName As String, _

                     ByVal strArtWorkFileName As String)

        Try

            ' Keep the original information to know what gets changed.

            m_strAlbumTitle = strAlbumTitle

            m_strArtistName = strArtistName

            m_strArtWorkFileName = strArtWorkFileName

            ' Show the fields to the user.

            Me.txtTitle.Text = strAlbumTitle

            Me.txtArtistName.Text = strArtistName

            ' If an art work file has been specified, show the artwork.

            If strArtWorkFileName.Length > 0 Then

                ' Make sure the file exists.

                If IO.File.Exists(strArtWorkFileName) Then

                    picArtWork.Image = Image.FromFile(strArtWorkFileName)

                Else

                MessageBox.Show("The artwork file " & strArtWorkFileName & _

" was not found.", "Album Catalog", MessageBoxButtons.OK, _     MessageBoxIcon.Information)

End If

            End If

        Catch ex As Exception

            MessageBox.Show("An error has occurred! Error: " & ex.Message, _

            "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

    End Sub

5. The next procedure will save changes made to an existing album or create a new album. If an album is being edited, its name will be stored in the module 

variable m_strAlbumTitle, and you’ll use this fact to determine whether a row is being added or edited.

Private Sub SaveAlbum()

        Dim cnADONetConnection As New OleDb.OleDbConnection

        Dim objCmd As New OleDb.OleDbCommand

        Dim objTrans As OleDb.OleDbTransaction

        Try

            ' Open the database.

            cnADONetConnection.ConnectionString = _

            "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & _

                     m_strDatabase

            cnADONetConnection.Open()

            ' Start a new transaction. A transaction is used to ensure that

' data is written to the disk immediately, and not held in cache.

            objTrans = cnADONetConnection.BeginTransaction

            ' Create a new Command object and attach the transaction to it.

            objCmd.Connection = cnADONetConnection

            objCmd.Transaction = objTrans

            ' Build the query text, depending on if an album is 

            ' being added or edited.

            If m_strAlbumTitle = "" Then

                ' Adding a new album.

         objCmd.CommandText = "INSERT INTO tblCDs (Title, ArtistName, " & _

               "ArtWorkFileName) " & _

                "VALUES ('" & txtTitle.Text & "','" & _

                txtArtistName.Text & "','" & m_strArtWorkFileName & "')"

            Else

                ' Editing an existing album.

      objCmd.CommandText = "UPDATE tblCDs SET Title = '" & txtTitle.Text & _

                    "', ArtistName = '" & txtArtistName.Text & "', " & _

                    "ArtWorkFileName = '" & m_strArtWorkFileName & _

                    "' WHERE Title = '" & m_strAlbumTitle & _

                    "' AND ArtistName = '" & m_strArtistName & "'"

            End If

            ' Execute the query.

            objCmd.ExecuteNonQuery()

            ' Commit the transaction to the database.

            objTrans.Commit()

        Catch ex As Exception

            ' If a transaction has been started, roll it back.

            If Not (objTrans Is Nothing) Then

                objTrans.Rollback()

            End If

            ' Display the error to the user.

            MessageBox.Show("An error has occured! Error: " & ex.Message, _

            "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

    End Sub
6. Return to the design view of the form and double-click the Art button to access its Click event. Add the following code:-

Try

            If ofdChooseArtwork.ShowDialog = DialogResult.OK Then

                ' Store the artwork file name.

                m_strArtWorkFileName = ofdChooseArtwork.FileName

                ' Show the artwork.

                picArtWork.Image = Image.FromFile(ofdChooseArtwork.FileName)

            End If

        Catch ex As Exception

    MessageBox.Show("An error has occured! Error: " & ex.Message, _       "CD Catalog", MessageBoxButtons.OK, MessageBoxIcon.Error)

        End Try

7. Return to the Design view, and double-click the OK button to access its Click event. Enter the following code;

Call SaveAlbum()

Me.Close()

8. Return to Design view again and double-click the Cancel button. Add the following code to its Click event:-
Me.Close()

Well done. Your application is complete. It only remains to test it.
This is a big project and there is a lot of detail here. You will need to be systematic in your study of this.

Run the program and try clicking on the New Album button on the toolbar. You’ll get a message saying that no database has been opened. This is light-years better than ‘throwing and exception’ and crashing the whole program. The catch-exception is easy to use and is a very powerful tool in the creation of user-friendly applications.

Try enlarging the form. The application is perfectly happy with this. This is the power of the Anchor property.
There is a lot of functionality in this little example, but the actual code is relatively easy to generate. This is productive application generation. To see one way in which this statement can be justified, think about what would be involved in doing this example from the beginning in Java....

I rest my case.

